Laboratory 3: Variational Calculus. Extremals for Integral Functionals

Euler-Lagrange Equation

Let consider the integral functional

$$J(y(x)) = \int_a^b L\left(x, y(x), \frac{d}{dx}y(x)\right) dx$$

and the conditions

$y(a) = \alpha$,

$y(b) = \beta$,

The extremals of the functional J are the solution of the Euler-Lagrange equation

$$L_y - \frac{d}{dx}(L_{y'})=0$$

Let's find the extremals for the following integral functional:

$$J(y(x)) = \int_0^{2\pi} \left(\frac{d}{dx}y(x)\right)^2 - y(x)^2 \, dx$$

with the conditions

$y(0) = 1$,

$y(2\pi) = 1$,

The lagrangian is defined by $L(x, u, v) = v^2 - u^2$

```
> L:=(x,u,v)->v^2-u^2;
```

$$L := (x, u, v) \to v^2 - u^2$$

In order to construct the Euler-Lagrange equation we need to calculate $\frac{\partial}{\partial u}L(x, u, v)$, $\frac{\partial}{\partial v}L(x, u, v)$ and evaluate them in the point ($x, y(x), \frac{d}{dx}y(x)$). To make the evaluation we need to use the partial derivative operator **`D[i](x,u,v)`**, which is the first partial derivative with respect to the i variable.

```
> D[1](L)(x,u,v);
```

$$0$$

```
> D[2](L)(x,u,v);
```

$$-2u$$

```
> D[3](L)(x,u,v);
```

$$2v$$

For our example we need to calculate **`D[2](L)(x,u,v)`**, **`D[3](L)(x,u,v)`**, evaluate them in the point $(x, y(x), \frac{d}{dx} y(x))$ and we have to derivate with respect to x the expression

`D[3](L)(x,y(x),diff(y(x),x))`

```
> EL_eq:=D[2](L)(x,y(x),diff(y(x),x))-
diff(D[3](L)(x,y(x),diff(y(x),x)),x)=0;
```

$$EL\_eq := -2\, y(x) - 2 \left(\frac{d^2}{dx^2} y(x) \right) = 0$$

Using **`dsolve`** command we can get the general solution

```
> dsolve(EL_eq,y(x));
```

$$y(x) = \_C1 \sin(x) + \_C2 \cos(x)$$

or we can use also the initial conditions and we get the expression of the extremal

```
> dsolve({EL_eq,y(0)=1,y(2*Pi)=1},y(x));
```

$$y(x) = \_C1 \sin(x) + \cos(x)$$

Notice, we get an infinite number of extremals for this example.

MAPLE has implemented this method in the package *VariationalCalculus*. For details see Help.

Euler-Lagrange System

The Euler-Lagrange system is used to find the extremals for integral functional that depend on several functions. Let's consider the case of two unknown functions:

$$J(y(x), z(x)) = \int_a^b L\left(x, y(x), z(x), \frac{d}{dx} y(x), \frac{d}{dx} z(x)\right) dx$$

and the initial conditions

$y(a) = \alpha_1$, $y(b) = \beta_1$

$z(a) = \alpha_2$, $z(b) = \beta_2$

In this case the Euler-Lagrange system has the following form

$$L_y - \frac{d}{dx} (L_{y} \; ,) = 0$$

$$L_z - \frac{d}{dx} (L_{z} \; ,) = 0$$

The lagrangian depends on 5 variables L=L(x,u1,u2,v1,v2).

Let's consider the following problem

$$J(y(x), z(x)) = \int_0^{\frac{\pi}{2}} \left(\frac{d}{dx} y(x) \right)^2 + \left(\frac{d}{dx} z(x) \right)^2 - 2\, y(x)\, z(x)\, dx$$

and the initial conditions

$$y(0)=2\ ,\ \ y\left(\frac{\pi}{2}\right)=e^{\left(\frac{\pi}{2}\right)}+1$$

$$z(0)=0\ ,\ \ z\left(\frac{\pi}{2}\right)=-e^{\left(\frac{\pi}{2}\right)}+1$$

First, we construct the lagrangian

```
> L:=(x,u1,u2,v1,v2)->v1^2+v2^2-2*u1*u2;
```

$$L := (x, u1, u2, v1, v2) \to v1^2+v2^2-2\,u1\,u2$$

To construct the Euler-Lagrange system we need to calculate $\frac{\partial}{\partial u1}\mathrm{L}(x, u1, u2, v1, v2)$, $\frac{\partial}{\partial v1}\mathrm{L}(x, u1, u2, v1, v2)$ and evaluate them in the point (x, $\mathrm{y}(x)$, $\mathrm{z}(x)$, $\frac{d}{dx}\mathrm{y}(x)$, $\frac{d}{dx}\mathrm{z}(x)$), for the first equation and $\frac{\partial}{\partial u2}\mathrm{L}(x, u1, u2, v1, v2)$, $\frac{\partial}{\partial v2}\mathrm{L}(x, u1, u2, v1, v2)$ evaluated in the point (x, $\mathrm{y}(x)$, $\mathrm{z}(x)$, $\frac{d}{dx}\mathrm{y}(x)$, $\frac{d}{dx}\mathrm{z}(x)$), for the second equation:

```
> eq1:=D[2](L)(x,y(x),z(x),diff(y(x),x),diff(z(x),x))-
diff(D[4](L)(x,y(x),z(x),diff(y(x),x),diff(z(x),x)),x)=0;
```

$$eq1 := -2\,\mathrm{z}(x)-2\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right)=0$$

```
> eq2:=D[3](L)(x,y(x),z(x),diff(y(x),x),diff(z(x),x))-
diff(D[5](L)(x,y(x),z(x),diff(y(x),x),diff(z(x),x)),x)=0;
```

$$eq2 := -2\,\mathrm{y}(x)-2\left(\frac{d^2}{dx^2}\mathrm{z}(x)\right)=0$$

We can get the general solution using **dsolve**

```
> dsolve({eq1,eq2},{y(x),z(x)});
```

$$\{\mathrm{z}(x)=-\_C1\,e^x-\_C2\,e^{(-x)}+\_C3\sin(x)+\_C4\cos(x),$$
$$\mathrm{y}(x)=\_C1\,e^x+\_C2\,e^{(-x)}+\_C3\sin(x)+\_C4\cos(x)\}$$

or we can use directly the initial conditions in order to get the expression of the extremal

```
> in_cond:=y(0)=2,y(Pi/2)=exp(Pi/2)+1,z(0)=0,z(Pi/2)=-exp(Pi/2)+1;
```

$$in\_cond := \mathrm{y}(0)=2,\ \mathrm{y}\left(\frac{\pi}{2}\right)=e^{\left(\frac{\pi}{2}\right)}+1,\ \mathrm{z}(0)=0,\ \mathrm{z}\left(\frac{\pi}{2}\right)=-e^{\left(\frac{\pi}{2}\right)}+1$$

```
> dsolve({eq1,eq2,in_cond},{y(x),z(x)});
```

$$\{\mathrm{y}(x)=e^x+\sin(x)+\cos(x),\ \mathrm{z}(x)=-e^x+\sin(x)+\cos(x)\}$$

Euler-Poisson Equation

The Euler-Poisson equation is used to find the extremals for integral functional that depend on higher-order derivatives of the unknown function. Let's consider the case when higher order is 2

$$J(y(x)) = \int_a^b L\left(x, y(x), \frac{d}{dx}y(x), \frac{d^2}{dx^2}y(x)\right) dx$$

and the initial conditions

$y(a) = \alpha_1$, $y(b) = \beta_1$

$y\,'(a) = \alpha_2$, $y\,'(b) = \beta_2$

In this case the Euler-Poisson equation has the following form

$$L_y - \frac{d}{dx}(L_{y'}) + \frac{d^2}{dx^2}(L_{y''}) = 0$$

The lagrangian depends on 4 variables L=L(x,u,v,w).

Let's consider the following problem

$$J(y(x)) = \int_0^1 4\left(\frac{d^2}{dx^2}y(x)\right)^2 + \left(\frac{d}{dx}y(x)\right)^2 + x\,y(x)\,dx$$

and the initial conditions

$$y(0) = -1 \text{ , } y(1) = e^{\left(\frac{1}{2}\right)} - e^{\left(-\frac{1}{2}\right)} - \frac{1}{12}$$

$$y\,'(0) = 2, y\,'(1) = \frac{1\,e^{\left(\frac{1}{2}\right)}}{2} + \frac{1\,e^{\left(-\frac{1}{2}\right)}}{2} + \frac{3}{4}$$

First, we construct the lagrangian

```
> L:=(x,u,v,w)->4*w^2+v^2-x*u;
```

$$L := (x, u, v, w) \rightarrow 4\,w^2 + v^2 - x\,u$$

To construct the Euler-Poisson equation we need to calculate $\frac{\partial}{\partial u}L(x, u, v, w)$, $\frac{\partial}{\partial v}L(x, u, v, w)$, $\frac{\partial}{\partial w}L(x, u, v, w)$ and evaluate them in the point ($x, y(x), \frac{d}{dx}y(x), \frac{d^2}{dx^2}y(x)$), for

$\frac{\partial}{\partial u}L(x, u, v, w)$ evaluated in the point ($x, y(x), \frac{d}{dx}y(x), \frac{d^2}{dx^2}y(x)$),

expression $\frac{\partial}{\partial v}L(x, u, v, w)$ evaluated in the point ($x, y(x), \frac{d}{dx}y(x), \frac{d^2}{dx^2}y(x)$) derivated with respect to x

expression $\frac{\partial}{\partial w}L(x, u, v, w)$ evaluated in the point ($x, y(x), \frac{d}{dx}y(x), \frac{d^2}{dx^2}y(x)$) derivated two times with respect to x

```
> EP_eq:=D[2](L)(x,y(x),diff(y(x),x),diff(y(x),x$2))-diff(D[3](L)(x,y(x),diff(y(x),x),diff(y(x),x$2)),x)+diff(D[4](L)(x,y(x),diff(y(x),x),diff(y(x),x$2)),x$2)=0;
```

$$EP\_eq := -x - 2\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right) + 8\left(\frac{d^4}{dx^4}\mathrm{y}(x)\right) = 0$$

We can get the general solution

```
> dsolve(EP_eq,y(x));
```

$$\mathrm{y}(x) = 4\,\mathbf{e}^{\left(\frac{x}{2}\right)}\,\_C2 + 4\,\mathbf{e}^{\left(-\frac{x}{2}\right)}\,\_C1 - \frac{x^3}{12} + \_C3\,x + \_C4$$

or, using the initial condition, we can get the extremal

```
> in_cond:=y(0)=-1,y(1)=exp(1/2)-exp(-1/2)-1/12,D(y)(0)=2,D(y)(1)=1/2*exp(1/2)+1/2*exp(-1/2)+3/4;
```

$in\_cond :=$

$$\mathrm{y}(0) = -1, \mathrm{y}(1) = \mathbf{e}^{(1/2)} - \mathbf{e}^{(-1/2)} - \frac{1}{12}, \mathrm{D}(y)(0) = 2, \mathrm{D}(y)(1) = \frac{1}{2}\mathbf{e}^{(1/2)} + \frac{1}{2}\mathbf{e}^{(-1/2)} + \frac{3}{4}$$

```
> dsolve({EP_eq,in_cond},y(x));
```

$$\mathrm{y}(x) = \mathbf{e}^{\left(\frac{x}{2}\right)} - \mathbf{e}^{\left(-\frac{x}{2}\right)} - \frac{x^3}{12} + x - 1$$